Bessel_6P_VCF

```
*dsauersanjose@aol.com  www.idea2ic.com  5/10/08
*
*    R0A= 1/A0A                   R0B= 1/A0B                 R0C= 1/A0C
*   _/\  /\_                 _/\  /\__                    _/\  /\_
*   |  \/    |    R3A= 1/A0A  |  \/    |       R3B= 1/A0B |   \/   |  R3C= 1/A0C
*   |     __|_/\  /\  /\_____|______  |_/\  /\  /\_______|______  |_/\  /\  /\_____________
*   |     |     \/  \/          LPA|  |   \/  \/            LPB|  |   \/  \/            LPC|
*  _|_    |    R2A= 1/A1A          |  |    R2B= 1/A1B          |  |    R2C= 1/A1C          |
* /VIN\   |_/\  /\  /\_____        |  |_/\  /\  /\_____        |  |_/\  /\  /\_____        |
* \___/   |   \/  \/       |       |  |   \/  \/       |       |  |   \/  \/       |       |
*   |     |   R1A= 1       |       |  |   R1  = 1      |       |  |   R1C= 1       |       |
*  _|_    |_/\  /\_        |BPA    |  |_/\  /\_        | BPB   |  |_/\  /\_        | BPC   |
*  ///    |   \/   |       |       |  |   \/   |       |       |  |   \/   |       |       |
*         | |\  HPA|XONES1A|XONES2A|  | |\  HPB|XONES1B|XONES2B|  | |\  HPC|XONES1C|XONES2C|
*         |_|-\    | _____ | _____ |  |_|-\    | _____ | _____ |  |_|-\    | _____ | _____ |
*       SUMA|  \___|_|1/s|_|_|1/s|_|    |  \___|_|1/s|_|_|1/s|_|    |  \___|_|1/s|_|_|1/s|_|
*           |  /     |___|   |___| SUMB |  /     |___|   |___| SUMC |  /     |___|   |___|
*         __|+/ OPAA  _|_     _|_     __|+/ OPAB  _|_     _|_     __|+/ OPAC  _|_     _|_
*        |  |/       \ /     \ /    |  |/       \ /     \ /    |  |/       \ /     \ /
*       _|_            V       V    _|_            V       V    _|_            V       V
*       ///              VBIAS      ///              VBIAS      ///              VBIAS
*       Bessel terms
*       (s^2 +1.861s +3.63)*(s^2 +2.76s +2.85)*(s^2 +3.143s + 2.57)
.OPTIONS  GMIN=1e-18    METHOD=gear   srcsteps = 1  gminsteps = 1
*======== ====== ====== ====== ====== ====== ====== ====== ======
V_IN      VIN     0       PULSE( -.5 .5 1u 1u  1u  10m 20m ) AC = 1
R0A       VIN     SUMA    .333
R1A       SUMA    HPA     1
R2A       SUMA    BPA     .54
R3A       SUMA    LPA     .333
XOPA1A    SUMA            HPA     OPA
XONES1A   HPA     BPA     VBIAS   ONE_S
XONES2A   BPA     LPA     VBIAS   ONE_S
R0B       LPA     SUMB    .35
R1B       SUMB    HPB     1
R2B       SUMB    BPB     .362
R3B       SUMB    LPB     .35
XOPA1B    SUMB            HPB     OPA
XONES1B   HPB     BPB     VBIAS   ONE_S
XONES2B   BPB     LPB     VBIAS   ONE_S
R0C       LPB     SUMC    .389
R1C       SUMC    HPC     1
R2C       SUMC    BPC     .318
R3C       SUMC    LPC     .389
XOPA1C    SUMC            HPC     OPA
XONES1C   HPC     BPC     VBIAS   ONE_S
XONES2C   BPC     LPC     VBIAS   ONE_S

E_INVERT  OUT0    0       LPC     0    -1
VBIAS     VBIAS   0       DC      1


*#0 =========A_Bessel_is_Best_for_Low_Phase_Distortion====
.control
set          pensize =    2
*#1 =========SET_VCF_VBIAS_TO_1=====================
set          outfile0 =              "Bessel_6P_VCF_1.txt"
tran         1m      40m      0   10m
run
plot         vin out0  title Vbias_1

let          saveData = 1
if           (saveData>0)
echo         "VpwlA  OUTA 0 PWL("    >$outfile0
let          NoOfTime    =    length(time)
echo        "Number of points is $&NoOfTime "
let          n = 0
repeat       $&NoOfTime
let          timestep = time[n] - time[n-1]
let          timme =  time[n]
let          vout   =  out0[n]
if           (timestep > 1u)
echo         "+ $&timme $&vout "   >> $outfile0
endif
let          n = n+1
end
echo        "+ )"                  >>$outfile0
endif
ac           dec     50     1    10000
run
plot         db(out0) ylimit -20 0   title Vbias_1
```

```
*#2 ============SET_VCF_VBIAS_TO_100m================
alter         vbias dc = .1
set           outfile1 =              "Bessel_6P_VCF_100m.txt"
tran          1m     40m      0  10m
run
plot          vin out0  title Vbias_100m

let           saveData = 1
if            (saveData>0)
echo          "VpwlB  OUTB 0 PWL("    >$outfile1
let           NoOfTime   =   length(time)
echo         "Number of points is $&NoOfTime "
let           n = 1
repeat        $&NoOfTime
let           timestep = time[n] - time[n-1]
let           timme =  time[n]
let           vout  =  out0[n]
if            (timestep > 1u)
echo          "+ $&timme $&vout "   >> $outfile1
endif
let           n = n+1
end
echo         "+ )"                   >>$outfile1
endif
ac            dec    50     1   10000
run
plot          db(out0) ylimit -20 0   title Vbias_100m

*#3 ============SET_VCF_VBIAS_TO_30m================
alter         vbias dc = .03
set           outfile2 =              "Bessel_6P_VCF_30m.txt"
tran          1m     40m      0  10m
run
plot          vin out0  title Vbias_30m

let           saveData = 1
if            (saveData>0)
echo          "VpwlC  OUTC 0 PWL("    >$outfile2
let           NoOfTime   =   length(time)
echo         "Number of points is $&NoOfTime "
let           n = 0
repeat        $&NoOfTime
let           timestep = time[n] - time[n-1]
let           timme =  time[n]
let           vout  =  out0[n]
if            (timestep > 1u)
echo          "+ $&timme $&vout "   >> $outfile2
endif
let           n = n+1
end
echo         "+ )"                   >>$outfile2
endif
ac            dec    50     1   10000
run
plot          db(out0) ylimit -20 0   title Vbias_30m

.endc
```

```
*==================OP_AMP======================================
.SUBCKT       OPA    IN     OUT
EGN1          OUT    0      IN     0     -10k
.ends
*                                          .SUBCKT  OPA  IN  OUT
*                           ____
*                      ____|OUT |
*                     _|_  |____|          |\
*       ____         /   \          ____   | \
*      | IN |_______|     |        | IN |__|-  \
*      |____|       | EGN1|  ---->  |____|  |    \___|OUT |
*                   |-10K |  ---->          |    /   |____|
*                  _|\___/                  |+  /
*                 |    |                 _|_ | /  OPA
*                 |____|                 ///
*                     _|_
*                     ///
*
*==================GM_C======================================
.SUBCKT       ONE_S  IN     OUT    VB
BGM           1      0      i = -10*v(IN)*v(VB)
C1            1      0      1m
EBUF          OUT    0      1      0      1
.ends
*
*   i = -10*v(IN)*v(VBIAS)                 ______|OUT |     .SUBCKT   ONE_S   IN    OUT   VBIAS
*                                          _|_   |____|
*  | VB |_    __|___________|__    /   \               | IN |_  ________  _|OUT |
*  |____| |_/   \         1    |  | EBUF|  ---->       |____| || 1/s    ||  |____|
*          \/ \/        _|_   _|  |  +1 |  ---->              ||________|
*  | IN |___/\_/\       ___   |    \___/                           |
*  |____|   \___/   C1 |      |_____|                            | VB |
*                                                                |____|
*      BGM   ///       \\\        ///
*

.end

=====================END_OF_SPICE=======================
```

A Bessel is designed for Low Phase Distortion.
A Voltage controlled Bessel will be able
to vary its frequency response while maintaining
Low Phase Distortion.

The 1/s block is gm_C block with gm a function of
input voltage VB. If all the 1/s blocks are varied
by a common VBIAS voltage, then frequency will vary
but the filter will still remain a 6pole bessel filter.

```
*#1 =========SET_VCF_VBIAS_TO_1====================
set           outfile0 =          "Bessel_6P_VCF_1.txt"
VBIAS         VBIAS   0      DC      1
```



The fundamental is a square wave at 25Hz.
With VBIAS set to 1V, the filter's bandwidth
is a little above 1kHz. Thats means the 40th

harmonic of the 25Hz is the 3dB point.



The output waveform is almost the same as the input waveform

```
*#2 =========SET_VCF_VBIAS_TO_100m================
alter          vbias dc = .1
set            outfile1 =              "Bessel_6P_VCF_100m.txt"
```

When VBIAS is set now to 100m, the bandwidth of the 6pole filter is now ten times smaller.





There is now about a 3 msec delay between the input and output waveforms. The 3dB point is now about at the 4th harmonic of 25Hz. The output waveform is a square wave with less sharp edges.

```
*#3 =========SET_VCF_VBIAS_TO_30m================
alter         vbias dc = .03
set           outfile2 =              "Bessel_6P_VCF_30m.txt"
```

Setting VBIAS down to 30m means all the odd harmonics are now attentuated.





The input to output delay has now increased to about 9msec and the output waveform is now close to being a pure 25Hz sinewave.